```
> od;
> M;
```

$$\begin{bmatrix} 0 & -c\,k & \sin(k\,L) & -\sin(k\,L)\,k^2 \\ 1 & -EJ\,k^2 & \cos(k\,L) & -\cos(k\,L)\,k^2 \\ 0 & -c & L & 0 \\ 1 & 0 & 1 & 0 \end{bmatrix}$$

```
> Dtr:=subs(EJ=b*c*L,L=t/k,Determinant(M));
```

$$Dtr := -c\sin(t)\,k^2 + c\,k^2\cos(t)\,t - b\,c\,t^2\,k^2\sin(t)$$

```
> b:=solve(Dtr,b);
```

$$b := -\frac{\sin(t) - \cos(t)\,t}{t^2\sin(t)}$$

```
> t:=fsolve(b-1,t=0.01..4);
```

$$t := 3.405608031$$

```
> mu:=evalf(Pi/t);#Коэффициент приведения длины
```

$$\mu := 0.9224762877$$

11.20. Устойчивость неоднородных стержней

В программе 34 реализован приближенный энергетический метод определения критической нагрузки сжатого стержня.

Заметим, что подстановку значений `x` в выражение `v` можно выполнить либо с помощью `subs(x=0,v)`, либо `eval(v,[x=0])`. В обоих случаях выражение `v` не меняется, и в него можно опять подставляет другие значения `x`.

В качестве примера решена задача на с. 251.

Программа 34

```
> restart;
> with(LinearAlgebra):
> a1:=1.5: a2:=1: k1:=1: k2:=1: c:=1/2:
  Задаем форму прогиба
> v:=sum(C[i]*(x/L)^i,i=0..6):
  Заделка в левой опоре
> eq[1]:=subs(x=0,v):
> eq[2]:=subs(x=0,diff(v,x)):
  Промежуточная шарнирная опора
> eq[3]:=subs(x=L*c,v):
  Шарнир на правой опоре
> eq[4]:=subs(x=L,v):
> eq[5]:=subs(x=L,diff(v,x$2)):
```

Определяем коэффициенты

```
> s:=solve({seq(eq[i],i=1..5)},{seq(C[i],i=0..4)}):
> assign(s):
```

Энергия изгиба

```
> J1:=int(a1*EJ*diff(v,x$2)^2,x=0..c*L)
>     +int(a2*EJ*diff(v,x$2)^2,x=c*L..L):
```

Работа сжимающих сил

```
> J2:=k1*int(diff(v,x)^2,x=0..c*L)
>     +k2*int(diff(v,x)^2,x=0..L):
> eqa:=simplify(evalf(P*J2-J1)*L^3):
```

Минимизируем критическую силу по C[5] и C[6]

```
> for i to 2 do
> s1[i]:=diff(eqa,C[i+4]);
> od:
> M:=Matrix(2):
> for i to 2 do
>   for j to 2 do
>     M[j,i]:=coeff(s1[i],C[j+4]);
>   od;
> od;
> M;
> solve(Determinant(M),P);
```

$$\begin{bmatrix} -0.6383928\,EJ + 0.01206289\,P\,L^2 & -2.006696\,EJ + 0.03853937\,P\,L^2 \\ -2.006696\,EJ + 0.03853937\,P\,L^2 & -6.388951\,EJ + 0.1238459\,P\,L^2 \end{bmatrix}$$

$$\frac{117.4238105\,EJ}{L^2},\ \frac{50.98540635\,EJ}{L^2}$$

Точное решение задачи об устойчивости стержня переменного сечения получаем из условия существования ненулевого решения дифференциального уравнения продольного изгиба.

Оператор `assume` использован для того, чтобы упростить выражение $L\sqrt{\dfrac{\xi^2}{L^2}}$, получающееся в результате решения задачи. Очевидно, Maple не может упростить результат до ξ, так как в зависимости от знаков L и ξ ответ имеет вид $\pm\xi$. Именно поэтому оператором `assume` мы назначаем знаки необходимым величинам. Аналогичный случай рассмотрен на с. 304.

Программа 35

```
> restart;
> k1:=sqrt(2*P/a1/EJ):k2:=sqrt(P/a2/EJ):
```
EJ1=1.5EJ, EJ2=EJ
```
> a1:=1.5: a2:=1:
```
Решение дифф.уравнения 4-го порядка ищем в виде
```
> y1:=C[1]*sin(k1*x)+C[2]*cos(k1*x)+C[3]*x+C[4]:
> y2:=C[5]*sin(k2*x)+C[6]*cos(k2*x)+C[7]*x+C[8]:
```
На левой опоре (заделка)
```
> eq[1]:=subs(x=0,y1):
> eq[2]:=subs(x=0,diff(y1,x)):
```
На промежуточной шарнирной опоре
```
> eq[3]:=subs(x=L/2,diff(y1,x)-diff(y2,x)):
> eq[4]:=subs(x=L/2,y1):
> eq[5]:=subs(x=L/2,y2):
> eq[6]:=subs(x=L/2,diff(y1,x$2)-diff(y2,x$2)):
```
На правой опоре (шарнир)
```
> eq[7]:=subs(x=L,y2):
> eq[8]:=subs(x=L,diff(y2,x$2)):
> with(LinearAlgebra):
```
Формируем матрицу системы для C[1],...,C[8]
```
> M:=Matrix(8):
> for i to 8 do
>  for j to 8 do
>   M[j,i]:=coeff(eq[i],C[j]);
>  od;
> od;
> assume(xi>0,L>0);
> U:=simplify(subs(P=xi^2*EJ/L^2,
>                  simplify(Determinant(M),trig)))*L^4:
> xi:=fsolve(U,xi=0.1..7.5);
> P1:=xi^2*EJ/L^2;
```
$$\xi := 6.898324570$$
$$P1 := \frac{47.58688187\,EJ}{L^{\sim 2}}$$